

ЎЗБЕКИСТОН РЕСПУБЛИКАСИ
ОЛИЙ ВА ЎРТА МАХСУС ТАЪЛИМ ВАЗИРЛИГИ
АНДИЖОН ДАВЛАТ УНИВЕРСИТЕТИ

А.А.Медатов

Информатикадан олимпиада масалалари

Андижон - 2011

А.А.Медатов. Информатикадан олимпиада масалалари. Методик қўлланма.-
Андижон,АДУ, 2011.-43 б.

Методик қўлланма ўрта мактаб ўқувчилари, академик лицей ва касб-хунар коллежлари талабалари учун мўлжалланган бўлиб, информатика фани бўйича вилоят олимпиадаларида қўлланилган масалалар ва мураккаб масалаларни ечишга мўлжалланган содда масалалардан ташкил топган. Ҳар бир масалага изох берилиб ечимлари Бейсик дастурлаш тилида ёритилган.

Масъул муҳаррир: т.ф.д., проф.Р.Алиев

Такризчилар: т.ф.н. О.А.Маматқулов, ф.-м.ф.н. М.Мирзаева

Методик қўлланма Заҳириддин Муҳаммад Бобур номидаги Андижон давлат университети информатика кафедрасида кўриб ўтилган ва чоп этишга тавсия қилинган.
20011 йил 20 майдаги йиғилиш баённомаси.

Андижон - 2011

Мундарижа

I. Бутун сонли арифметика	4
II. Бир ўлчовли жадвал катталиклар	12
III. Икки ўлчовли жадвал катталиклар	16
IV. Кетма-кет яқинлашиш	22
V. Максимум ва минимумларга доир масалалар	24
VI. Геометрик масалалар	26
VII. Белгили катталиклар билан ишлаш функцияларига оид масалалар	33
VIII. Паскаль учбурчаги	40
IX. Календар	42

I. Бутун сонли арифметика

1. Қийматлари ҳақиқий сонлардан иборат бўлган иккита a ва b ўзгарувчи берилган. Учинчи ўзгарувчидан фойдаланмаган ҳолда бу ўзгарувчиларнинг қийматларини алмаштиринг.

Ечиш. Ўзгарувчиларнинг қийматлари сонлардан иборат бўлганлиги учун уларнинг қийматларини қуйидаги кўринишда алмаштириш мумкин:

$a=a+b;$

$b=a-b;$

$a=a-b;$

2. 3 та чақа орасидан сохтасини бир марта тортиш ёрдамида аниқланг.

Ечиш: Иккита чақа тортилади, уларнинг оғирлиги бир-бирига тенг бўлса, учинчи чақа сохта, уларнинг бири иккинчисидан енгил бўлса, енгили сохта.

3. Ихтиёрий $N>7$ бутун сон учун $3x+5y=N$ шартни қаноатлантирувчи барча бутун x ва y сонлар жуфтлигини топинг.

Ечиш. N сонини 5 га бўлиб унинг бутун қисми K га эга бўламиз. y нинг $0 \leq y \leq K$ оралиқдаги барча бутун қийматлари учун $N-5y$ сонини 3 га кетма-кет бўлиб чиқамиз. Хосил бўлган сонлар ичида қолдиқсиз бўлинганлари x нинг мос қийматлари бўлади.

```
10 INPUT "n=";N
20 K=INT(N/5)
30 FOR Y=1 TO K
40 IF (N-5*Y) MOD 3 <> 0 THEN 60
50 PRINT"x=";INT(N-5*Y)/3,"y=";Y
60 NEXT Y
70 END
```

4. 2, 3, 4, 5 ва 6 ларга бўлганда 1 қолдиқ қолиб, 7 га қолдиқсиз бўлинадиган биринчи натурал сонни топинг.

Ечиш: 7 га қаррали бўлган сонлар ичидан 2, 3, 4, 5, 6 ларга бўлганда 1 қолдиқ қоладиган биринчи натурал сон топилади.

```
10 I=1
20 N=7*I:S=0
30 FOR J=2 TO 6
40 IF N MOD J = 1 THEN S=S+1
50 NEXT J
60 IF S=5 THEN 80
70 I=I+1: GOTO 20
80 PRINT"n=";N
90 END
```

5. Рақамлари бир-бирига тенг бўлмаган шундай тўрт хонали сонларни топингки, улардаги тоқ ўринда турган рақамлар йиғиндиси жуфт ўринда турган рақамлар йиғиндисига тенг бўлсин.

Ечиш. Ҳар қандай тўрт хонали сонни $\overline{abcd}=1000*a+100*b+10*c+d$ кўринишда ёзиш мумкин. Шартга кўра $a \neq b$, $a \neq c$, $a \neq d$, $b \neq c$, $b \neq d$, $c \neq d$, $a+c=b+d$ бўлиши керак. Шунинг учун дастур қуйидагича бўлади:

```

10 FOR I=1 TO 9
20 FOR J=1 TO 9
30 FOR K=1 TO 9
40 FOR L=1 TO 9
50 IF I<>J AND I<>K AND I<>L AND J<>K AND J<>L AND K<>L AND I+K=J+L THEN
PRINT 1000*I+100*J+10*K+L;
60 NEXT L,K,J,I
70 END

```

6. А шаҳардан В шаҳарга 20 та турли хилдаги юкларни автоуловларда ташиш керак. Юклар 1, 2, 3, . . . каби номерланган бўлиб, уларни бўлиб юклаш мукин эмас ва бир хил кетма-кетликда юклаш талаб қилинади. Ҳар бир автоуловнинг юк ташиш имконияти 15 тоннадан ошмаса, юк ташиш учун нечта автоулов керак бўлади.

Ечиш: $a(i)$ -юкларнинг вазни, k -автоулов рақами, i -навбатдаги юк рақами, s орқали k -автоуловдаги юкнинг массаси бўлсин. У ҳолда k – автоуловга навбатдаги юк юкланади. Агарда унга кейинги юк ҳам жойлашса у ҳам юкланади, акс ҳолда k ни бирга ортириб кейинги юк $k+1$ автоуловга жойлаштирилади.

```

10 DIM A(20)
20 FOR I=1 TO 20
30 PRINT "a(";I;")=";
40 INPUT A(I)
50 NEXT I
60 S=0:I=1:K=1
70 WHILE I<=20
90 IF S+A(I)<=15 THEN S=S+A(I):I=I+1:GOTO 110
100 K=K+1:S=0
110 WEND
120 PRINT K;" ta yuk mashinasi kerak bo'ladi"
130 END

```

7. 1 да 2000 гача бўлган сонлар ичидан рақамлари кубларининг йиғиндиси шу сонга тенг бўладиган сонларни аниқлаш дастури тузилсин.

Ечиш: Ушбу масалани ечишда 1 дан бошлаб 2000 гача бўлган ҳар бир сонни ўнг томонидан бошлаб, яъни бирликлар хонасидан бошлаб барча рақамларини аниқлаш зарур бўлади. Бунинг учун Текшириляётган X сони P сони билан алмаштирилади. P сонини 10 га бўлиб қолдиқ ҳад N ($N = P \text{ MOD } 10$) ва P ни тўлиқсиз бўлинма ($P = \text{INT}(P/10)$) билан алмаштирилади. +олдиқ ҳадни кубга кўтариб йиғинди S га қўшилади. P нолдан катта бўлса, жараён давом эттирилади. Жараён тугагач текшириляётган X сони билан йиғинди S солиштирилади. Бу сонлар бир-бирига тенг бўлса, натижа экранга чиқарилади. Акс ҳолда навбатдаги сон текширилади.

```

10 FOR X=1 TO 2000
20 S=0:P=X
30 WHILE P>0
40 N = P MOD 10
50 P=INT(P/10)
60 S=S+N^3
70 WEND
80 IF X=S THEN PRINT S

```

90 NEXT

8. Агар
AA
+
BB

SBS бўлса A, B, S рақамларни топиш дастурини тузинг.

Ечиш: Маълумки, ўн асосли саноқ системасидаги ҳар қандай сонни $\dots abcd = \dots + 1000 * a + 100 * b + 10 * c + d$ кўринишда ифодалаш мумкин. Шунинг учун берилган икки хонали соннинг охири рақамини топиш учун шу сонни 10 га бўлиб қолдик қисмини ажратиш кифоя, яъни $b = \overline{ab} - \text{int}(\overline{ab}/10)$ га тенг бўлади. Дастур тузишда A ва B рақамлар бўйича цикл очилаб, ҳосил бўладиган уч хонали соннинг ҳар бир рақами S ва B лар билан солиштирилади.

```
10 T=0
20 FOR A=1 TO 9
30 FOR B=1 TO 9
40 N=10*A+A+10*B+B
50 K=INT(N/100)
60 L=INT((N-K*100)/10)
70 S=N-K*100-L*10
80 IF S<>K OR L<>B THEN 110
90 PRINT "a=";A,"b=";B,"s=";S
100 T=T+1
110 NEXT B
120 NEXT A
130 IF T=0 THEN PRINT"Bunday yig'indi mavjud emas"
140 END
```

9. Агар
AAA
+
BBB

SBBS бўлса A, B, S рақамларни топиш дастурини тузинг.

Ечиш: Бу масала ҳам ю=оридаги масала каби ечилади.

```
10 T=0
20 FOR A=1 TO 9
30 FOR B=1 TO 9
40 N=100*A+10*A+A+100*B+10*B+B
50 K=INT(N/1000)
60 L=INT((N-K*1000)/100)
70 M=INT((N-K*1000-L*100)/10)
80 S=N-K*1000-L*100-M*10
90 IF S<>K OR L<>M OR M<>B THEN 120
100 PRINT "a=";A,"b=";B,"s=";S
110 T=T+1
120 NEXT B
```

```

130 NEXT A
140 IF T=0 THEN PRINT"Bunday yig'indi mavjud emas"
150 END

```

10. Натурал сон мукаммал дейилади, агар ўзидан ташқари ҳамма бўлувчиларининг йиғиндисига тенг бўлса. Масалан, $6=1+2+3$, яъни 6-мукаммал; $8>1+2+4$, 8- мукаммал эмас. Берилган N натурал сонидан кичик барча мукаммал сонларни аниқланг.

Ечиш: Берилган I сонини унинг ярмигача бўлган барча натурал сонларга кетма-кет бўлиб чиқилади. Шу сонлар ичидан I га бўлинадиган барча сонлар қўшилиб I билан солиштирилади. Йиғинди билан I бир-бирига тенг бўлса I сони мукаммал бўлади. Шу тариқа барча N гача бўлган сонлар текширилади.

```

10 INPUT "n=";N
20 K=0
30 FOR I=2 TO N
40 S=1
50 FOR J=2 TO INT(I/2)
60 IF I MOD J = 0 THEN S=S+J
70 NEXT J
80 IF S=I THEN PRINT I;" mukammal son":K=K+1
90 NEXT I
100 IF K=0 THEN PRINT N;" gacha bo'lgan sonlar ichida mukammal son yo'q"
110 END

```

11. Берилган сон квадратларининг охириги рақамлари шу сонга тенг бўлса бундай сонлар автоморф сонлар дейилади. Масалан, $6^2=36$, $25^2=625$, $76^2=5776$.
m ва n сонлар орасидаги автоморф сонлар сонини аниқлаш дастури тузилсин.

Ечиш: Автоморф сонларни аниқлашда текширилаётган сонлар 0 дан 9 гача бўлганда бу сонларнинг квадратларини 10 га бўлгандаги қолдиқ билан тенглаштириш, 10 дан 99 гача бўлганда бу сонларнинг квадратларини 100 га бўлгандаги қолдиқ билан тенглаштириш ва ҳ.к. кифоя. Бу сонлар билан қолдиқ сонлар тенг бўлса, бундай сонларни автоморф сонлар бўлади. Масалан, I=5 бўлганда $I^2=25$ бўлади. $25 \bmod 10 = 5$ бўлганлиги учун 5 автоморф сон бўлади.

```

10 INPUT "m,n ";M,N
20 A=10
30 FOR I=M TO N
40 WHILE I>=A
50 A=A*10
60 WEND
70 IF I^2 MOD A =I THEN PRINT I,I^2
80 NEXT I
90 END

```

12. N, M натурал сонлар берилган. Йиғиндиси M га тенг бўлган барча N дан кичик жуфт сонлар жуфтлигини аниқловчи дастур тузинг.

Ечиш: Масала шартига кўра M сонининг жуфт бўлиши аниқ. Агар M тоқ бўлса бундай сон йўқ деган жавоб олиниши керак. Масалани ечишда $N-1$ гача булган барча жуфт сонлар кетма-кет қўшилиб M сонига тенглаштирилади. Йиғинди M га тенг бўлса, қўшилувчилар жуфтлиги экранга чиқарилади.

Масалани мураккаблаштириш учун «рақамларининг йиғиндиси M га тенг бўлган барча N дан кичик жуфт сонларни аниқловчи дастур тузинг» деб ўзгартирилиши мумкин.

```
10 INPUT "n,m ";N,M
20 IF N<0 OR M<0 THEN PRINT"n va m larga musbat qiymatlar kiriting":GOTO 10
30 K=0
40 FOR I=2 TO N-1 STEP 2
50 FOR J=2 TO N-1 STEP 2
60 IF I+J=M THEN PRINT I,J ELSE 80
70 K=K+1
80 NEXT J,I
90 IF K=0 THEN PRINT"Bunday son yo'q"
100 END
```

13. N , M натурал сонлар берилган. Йиғиндиларининг квадрати M га тенг бўлган барча N дан кичик натурал сонлар жуфтлигини аниқланг.

Ечиш: Барча N дан кичик сонлар йиғиндисининг квадрати M билан солиштирилади. Агар йиғинди квадрати M га тенг бўлса шу сонлар экранга чиқарилади. Бу масалани ечишда ичма-ич жойлашган циклнинг очилиши мақсадга мувофиқ.

```
10 INPUT "n,m ";N,M
20 K=0
30 FOR I=1 TO N-1
40 FOR J=I TO N-1
50 IF (I+J)^2<>M THEN 80
60 K=K+1
70 PRINT I,J
80 NEXT J,I
90 IF K=0 THEN PRINT"Bunday son yo'q"
100 END
```

14. Берилган натурал сонни туб кўпайтувчилар кўпайтмаси шаклида ёзинг.

Ечиш: Аввало берилган сонни туб ёки мураккаб эканлигини аниқлаш лозим (20-40 сатрлар). N натурал сони туб бўлиши учун u 2 дан бошлаб \sqrt{N} нинг бўтун қисмигача бўлган барча натурал сонларга бўлинмаслигини текшириш kifоя. Агар берилган сон туб бўлса, уни туб эканлигини эълон қилиб, жараёни тўхтатиш лозим (50 сатр).

Берилган сон мураккаб бўлса, унинг барча туб бўлувчиларини аниқлаб (90-120 сатрлар), фақат шу сонларгагина бўлиш зарур (130-150 сатрлар).

```
10 INPUT "n=";N
20 FOR I=2 TO INT(SQR(N))
30 IF N/I=INT(N/I) THEN 60
40 NEXT I
50 PRINT N;" tub son": GOTO 170
60 PRINT N;"=";
```

```

70 FOR I=2 TO INT(N/2)
80 K=0
90 FOR J=2 TO INT(SQR(I))
100 IF I/J=INT(I/J) THEN 160
110 NEXT J
120 IF N/I<>INT(N/I) AND K=0 THEN 160
130 IF N/I<>INT(N/I) THEN 150
140 N=N/I:K=K+1:GOTO 130
150 PRINT I;"^";K;"*";
160 NEXT I
170 END
RUN

```

15. 3^{1000} сонининг барча рақамларини экранга чиқариш дастури тузилсин.

Ечиш: Компьютер хотирасига сиғмайдиган катта сонларни ёзишда уларнинг рақамларини жадвал элементлари шаклида ифодалаб олиш қулай бўлади. $3^{1000}=(3^2)^{500}=9^{500}\approx 10^{500}$ бўлганлиги учун бу сонни ёзиш учун тахминан 500 тагача бўлган рақамлар қатнашади, яъни бу сон тахминан 500 хонали сондан иборат бўлади.

Шунинг учун 500 хонали жадвал эълон қилинди. Аввало жадвалга 3 сонини киритиб уни кетма-кет 3 га кўпатириб борилади, бунда албатта бирлар хонасидан ўнлар хонасига ва х.к. ўтишда разряддан-разрядга (R ўзгарувчи)ўтиш ҳисобга олинади.

Масалани тўла тушуниш учун 3^{12} ни ҳисоблаш жараёнини текшириб чиқиш тавсия этилади. 3^{12} нинг рақамларини экранга чиқариш учун $3^{12}=(3^2)^6=9^6\approx 10^6$ бўлганлигини ҳисобга олган ҳолда дастурдаги 1000 ўрнига 12, 500 ўрнига 6 сонини алмаштириш кифоя.

```

10 DIM A(500)
20 A(500)=3:R=0
30 FOR I=2 TO 1000
40 FOR J=500 TO 1 STEP -1
50 X=A(J)*3
60 A(J)=(X+R) MOD 10
70 R=INT((X+R)/10)
80 NEXT J,I
90 K=1
100 WHILE A(K)=0
110 K=K+1
120 WEND
130 FOR I=K TO 500
140 PRINT A(I);
150 NEXT
160 END

```

II. Бир ўлчовли жадвал катталиклар

1. Кетма-кетликнинг биринчи учта ҳади – 1,0,1 эканлиги маълум. Кейинги ҳар бир ҳад ўзидан олдинги учта ҳаднинг йиғиндисига тенг бўлса, кетма-кетликнинг биринчи N та ҳадини чиқарувчи дастур тузинг.

Ечиш: Жадвалнинг биринчи учта ҳади маълум бўлганлиги учун кейинги ҳадларини $A(I)=A(I-1)+A(I-2)+A(I-3)$ боғланиш орқали ифодаланиши мумкин.

```
10 INPUT N%
20 DIM A%(N%)
40 A%(1)=-1
50 A%(2)=0
60 A%(3)=1
70 FOR I=4 TO N%
80 A%(I)=A%(I-1)+A%(I-2)+A%(I-3)
90 NEXT I
100 FOR I=1 TO N%
110 PRINT A%(I); " ";
120 NEXT I
130 END
```

2. n натурал сони ва $a_1, a_2, a_3, \dots, a_n$ ҳақиқий сонлар кетма-кетлиги берилган. $a_1, a_2, a_3, \dots, a_n$ ичида кетма-кет келувчи бир хил элементларнинг энг кўп сонини ва шу ҳадни аниқлаш дастури тузилсин.

Ечиш: S -кетма кет келувчи бир хил элементларнинг энг кўп сони, K -кетма кет келувчи сонлар учун оралик ўзгарувчи, T -кетма кет келувчи элемент бўлсин. Кейинги сон кетма-кет аввалги сон билан солиштирилади. Агар бу сонлар бир-бирига тенг бўлмаса текшириш давом эттирилади, акс ҳолда K нинг қиймати биттага орттирилади. Цикл охирига етган ва $K>S$ бўлса $S=K:T=A(I):K=1$ алмаштиришлар қилиниб цикл давом эттирилади. Дастурнинг 100 сатри кетма-кет келувчи сонлар тугагандан кейин ҳам албатта $K<S$ шартнинг текширилиши зарурлигини англатади.

```
10 INPUT "Elementlar soni n=";N
20 IF N<1 THEN 10
30 DIM A(N)
40 FOR I=1 TO N
50 PRINT "a(";I;")=";
60 INPUT A(I)
70 NEXT I
80 S=1:K=1
90 FOR I=2 TO N
100 IF A(I)<>A(I-1) THEN 140
110 K=K+1
120 IF I=N AND K>S THEN 150
130 GOTO 160
140 IF K<S THEN 160
150 S=K:T=A(I):K=1
160 NEXT I
170 PRINT"Jadvalda ketma-ket keluvchi bir ";
180 PRINT" xil elementlarning eng ko'p soni ";S;" ta"
```

```
190 PRINT"Ketma-ket keluvchi sonlar ";T
200 END
```

3. n натурал сони ва $a_1, a_2, a_3, \dots, a_n$ бутун сонлар кетма-кетлиги берилган. $a_1, a_2, a_3, \dots, a_n$ кетма-кетликда учраган ҳар хил сонларни экранга чиқариш ва уларнинг сонини аниқловчи дастур тузилсин.

Ечиш: Кетма-кетликда ҳар хил сонлар сонини аниқлашда кейинги ҳаддан бошлаб аввалги ҳар бир ҳадни текшириш мақсадга мувофиқ. Шунинг учун ички цикл очилиб такрорланиш бор (t.b) ёки такрорланиш йўқ (t.y)лиги текширилади.

```
10 INPUT "Elementlar soni n=";N%
20 IF N%<1 THEN 10
30 DIM A%(N%)
40 FOR I=1 TO N%
50 PRINT "a%(";I;")=";
60 INPUT A%(I)
70 NEXT I
80 S=0
90 FOR I=2 TO N%
100 B$="t.y"
110 J=1
120 WHILE J<=I-1 AND B$="t.y"
130 IF A%(J)=A%(I) THEN B$="t.b" ELSE J=J+1
140 WEND
150 IF B$="t.y" THEN S=S+1
160 NEXT I
170 PRINT "Har xil sonlar soni ";S;" ta"
180 END
```

4. Бутун қийматли $A[1:N]$ жадвал берилган. A жадвалнинг барча элементларидан тузилган шундай $B[1:N]$ жадвал тузингки, ундаги барча манфий элементлар манфий бўлмаган элементлардан олдин келсин.

Ечиш: $A[1:N]$ жадвал ҳар бир ҳадининг ишораси текширилиб мусбат бўлса $B[1:N]$ жадвалнинг мос элементига ўтказилади, акс ҳолда $B[1:N]$ жадвалнинг биринчи элементидан бошлаб кетма-кет SWAP $B(J), B(I)$ фармойиши ёрдамида алмаштириб борилади.

```
10 INPUT "n=";N%
20 DIM A%(N%),B%(N%)
30 FOR I=1 TO N%
40 PRINT"a%(";I;")=";
50 INPUT A%(I)
60 NEXT I
65 J=0
70 FOR I=1 TO N%
80 IF A%(I)>0 THEN B%(I)=A%(I): GOTO 100
90 J=J+1:B%(I)=A%(I): SWAP B%(J),B%(I)
100 NEXT I
110 FOR I=1 TO N%
120 PRINT"b%(";I;")=";B%(I)
```

5. Берилган $X[1..N]$ массивни K элементга циклик суриш керак, яъни $X(I)$ элемент $X(I+K)$ элемент ўрнига ўтиши керак; $X(N)$ элементдан кейин $X(1)$ элемент келади. (Қўшимча массивдан фойдаланиш мумкин эмас).

Ечиш: Ихтиёрый жадвалнинг иккита элементининг ўринларини алмаштириш учун қўшимча ўзгарувчидан фойдаланилади, яъни j -элемент билан N -чи элементларнинг ўринларини алмаштириш учун $M=A(J)$: $A(J)=A(N)$: $A(N)=M$ кўринишдаги шакл алмаштиришларни бажариш етарли. Шунинг учун масалани ечишда ичма-ич жойлашган циклдан фойдаланган ҳолда жадвал элементларини суриш мумкин. Ташқи цикл бир марта ва ички цикл тўла бажарилганда жадвал элементларининг ҳаммаси бир хонадан сурилади. Шунинг учун ташқи цикл параметрининг охириги қиймати (K) суришлар сонини англатади.

```

10 INPUT "Elementlar soni N=";N
20 DIM A(N)
30 FOR I=1 TO N
40 PRINT"A(";I;")=";
50 INPUT A(I)
60 NEXT I
70 INPUT "Nechta elementga surish kerak?";K
80 IF K>N THEN K = K MOD N
90 FOR I=1 TO K
100 FOR J=1 TO N
110 M=A(J): A(J)=A(N): A(N)=M
120 NEXT J,I
130 FOR I=1 TO N
140 PRINT A(I);
150 NEXT I
160 END

```

6. n натурал сони, $a_1, a_2, a_3, \dots, a_n$ ва $b_1, b_2, b_3, \dots, b_n$ сонлар кетма-кетлиги берилган. Улардан шундай n та (a_i, b_j) жуфтлик тузиш керакки, (a_i, b_j) кўпайтмаларнинг йиғиндиси энг кичик бўлсин.

Ечиш: Маълумки, иккита N та элементли сонлар жуфтлиги $N*N$ та бўлади, яъни бундай сонлар жуфтлиги

$(a_1, b_1), (a_1, b_2), \dots, (a_1, b_n),$

$(a_2, b_1), (a_2, b_2), \dots, (a_2, b_n),$

.....

$(a_n, b_1), (a_n, b_2), \dots, (a_n, b_n)$ дан иборат бўлади.

Масалани ечиш учун қўшимча $N*N$ та элементли жадвал очиб, хосил қилинган сонлар жуфтликларининг кўпайтмалари сақланади. Бу кўпайтмаларни ўсиб бориш тартибда саралаб биринчи N тасининг йиғиндиси олинади.

```

10 INPUT "n=";N
20 S=0
30 DIM A(N),B(N),K(N*N)
40 FOR I=1 TO N
50 PRINT "A(";I;"),B(";I;") larni kiriting ";
60 INPUT A(I),B(I)

```

```

70 NEXT I
80 L=1
90 FOR I=1 TO N
100 FOR J=1 TO N
110 K(L)=A(I)*B(J)
120 L=L+1
130 IF L>N*N THEN 150
140 NEXT J,I
150 FOR I=1 TO N
160 T=K(I)
170 FOR J=I TO N*N
180 IF T<K(J) THEN 210
190 T=K(J)
200 X=J
210 NEXT J
220 SWAP K(I),K(X)
230 S=S+K(I)
240 NEXT I
250 PRINT S
260 END

```

III. Икки ўлчовли жадвал катталиклар

1. $A[1:N]$ ва $B[1:M]$ бутун қийматли жадвал берилган. Бу жадваллар ўхшаш эканлигини текширувчи дастур тузинг. Иккита жадвал ўхшаш дейилади, агар бу жадвалларда учровчи сонлар тўплами бир хил бўлса.

Ечиш: Биринчи жадвалнинг ҳар бир элементи билан иккинчи жадвалнинг барча элементлари кетма-кет солиштириб чиқарилади. Биринчи жадвалнинг бирорта элементи иккинчи жадвалда қатнашмаса, жадваллар ўхшаш бўлмайди. Шунинг сингари иккинчи жадвалнинг ҳар бир элементи биринчи жадвалнинг барча элементлари билан ҳам солиштирилади, чунки иккинчи жадвалда биринчи жадвалда қатнашмаган элементларнинг ҳам бўлиб қолишини текшириб чиқиш зарур.

```

10 INPUT "n=";N
20 INPUT "m=";M
30 DIM A(N),B(M)
40 FOR I=1 TO N
50 PRINT"a(";I;")="";INPUT A(I)
60 NEXT I
70 FOR I=1 TO M
80 PRINT"b(";I;")="";INPUT B(I)
90 NEXT I
100 FOR I=1 TO N
110 K=0
120 FOR J=1 TO M
130 IF A(I)<>B(J) THEN 160
140 K=K+1
150 GOTO 180
160 NEXT J
170 IF K=0 THEN 300
180 NEXT I
190 FOR J=1 TO M

```

```

200 K=0
210 FOR I=1 TO N
220 IF A(I)<>B(J) THEN 250
230 K=K+1
240 GOTO 270
250 NEXT I
260 IF K=0 THEN 300
270 NEXT J
280 PRINT "Jadvallar o'xshash"
290 GOTO 310
300 PRINT "Jadvallar o'xshash emas"
310 END

```

2. Бутун қийматли $A(N,M)$ жадвал ҳар бир сатри элементлари модулларининг йиғиндисининг энг каттасини топувчи дастур тузинг.

Ечиш: Масалани ечишда ҳар бир сатр элементлари модулларининг йиғиндисини кетма-кет бир ўлчовли жадвал элементлари билан алмаштирилиши, сўнгра уларнинг максимумини топиш кифоя.

```

10 INPUT "Satrlar soni=";N
20 INPUT "Ustunlar soni=";M
30 DIM A(N,M), B(M)
40 FOR I=1 TO N
50 PRINT I;"-satr elementlarini kirining:"
60 FOR J=1 TO M
70 PRINT"a(";I;" ";J;" )=";
80 INPUT A(I,J)
90 NEXT J,I
100 FOR I=1 TO N
110 S=0
120 FOR J=1 TO M
130 S=S+ABS(A(I,J))
140 NEXT J
150 B(I)=S
160 NEXT I
165 S=B(1)
170 FOR I=2 TO M
180 IF B(I)>S THEN S=B(I)
190 NEXT I
200 PRINT"Eng kattasi=";S
210 END

```

3. Элементлари ҳақиқий сонлардан иборат бўлган тўғри тўртбурчакли $A[N,M]$ жадвал берилган. Жадвал сатрлари максимумларининг минимуми топилсин.

Ечиш: Жадвал ҳар бир сатрининг максимумларидан алоҳида жадвал тузилиши ва тузилган жадвал элементларининг минимуми топилади.

```

10 INPUT "Jadvaldagi satrlar soni N=";N
20 INPUT "Jadvaldagi ustunlar soni M=";M
30 DIM A(N,M), B(N)

```

```

40 FOR I=1 TO N
50 FOR J=1 TO M
60 PRINT "a(";I;",";J;")=";
70 INPUT A(I,J)
80 NEXT J
90 NEXT I
100 FOR I=1 TO N
110 K=A(I,1)
120 FOR J=2 TO M
130 IF A(I,J)>K THEN K=A(I,J)
140 NEXT J
150 B(I)=K
160 NEXT I
170 L=B(1)
180 FOR I=2 TO N
190 IF B(I)<L THEN L=B(I)
200 NEXT I
210 PRINT "Bunday kichik son ";L
220 END

```

4. Элементлари ҳақиқий сонлардан иборат булган тўғри бурчакли $A(N,M)$ жадвал берилган. Жадвал элементлари ичидан шундай K сонни топингки жадвалнинг ҳеч булмаганда битта сатридаги ҳамма элементлар K дан катта бўлмасин.

Ечиш: Бу масала ҳам аввалги масалага ўхшаш бўлиб жадвал сатрлари минимумларининг максимумини аниқлаш деган мазмунни билдиради.

```

10 INPUT N
20 INPUT M
30 DIM A(N,M)
40 FOR I=1 TO N
50 FOR J=1 TO M
60 PRINT "A(";I;",";J;")=";:INPUT A(I,J)
70 NEXT J,I
100 K=A(1,1)
110 FOR J=2 TO M
120 IF A(1,J)>K THEN K=A(1,J)
130 NEXT J
140 FOR I=2 TO N
150 L=A(I,1)
155 FOR J=2 TO M
160 IF A(I,J)>L THEN L=A(I,J)
170 NEXT J
180 IF L<K THEN K=L
190 NEXT I
200 PRINT "BUNDAY KICHIK SON-";K
210 END

```


5. N ўлчовли квадрат жадвал берилган. Бўялган соҳадаги элементлар қийматлари ичидан энг катта ва энг кичигини топинг.

Ечиш: Квадрат жадвалнинг бўялган соҳасидаги қийматларни аниқлаш учун ичма-ич жойлашган цикл очиб, ички цикл бир марта бажарилганидан сўнг ташқи цикл охириги қийматидан бирни айириш кифоя.

```


10 INPUT "n=";N
20 DIM A(N,N)
30 FOR I=1 TO N
40 FOR J=1 TO N
50 PRINT "a(";I;" ";J;")=";
60 INPUT A(I,J)
70 NEXT J
80 NEXT I
90 KAT=A(1,1):KICH=A(1,1)
100 FOR I=1 TO N
110 FOR J=I TO N
120 IF KAT<A(I,J) THEN KAT=A(I,J) ELSE KICH=A(I,J)
130 NEXT J
140 N=N-1
150 NEXT I
160 PRINT "Eng kattasi=";KAT
170 PRINT "eng kichigi=";KICH
180 END

```



6. N ўлчовли квадрат жадвал берилган. Бўялган соҳадаги элементлар ичидан энг катта ва энг кичигини топинг.

Ечиш: Бу масала ҳам аввалги масалага ўхшаш бўлиб, квадрат жадвалнинг бўялган соҳасидаги элементлар максимуми ва минимумини аниқлашни талаб қилади. Дастур тузиб кўриринг.



7. Бутун қийматли квадрат $A(N,N)$ жадвал берилган. Агар i -сатр элементлари ўсмайдиган бўлса, $b_i=1$, акс ҳолда 0 га тенг бўлган $b_1..b_n$ бирлар ва ноллар кетма-кетлигини тузинг.

Ечиш: Масала ечими содда бўлиб, $A(N,N)$ жадвалнинг ҳар бир сатри элементлари усмайдиган бўлса, яъни $A(I,J-1)<A(I,J)$ шарт бажарилмаса, $B(I)=1$ акс ҳолда $B(I)=0$ қийматлари берилади.

```

10 INPUT "n=";N
20 DIM A(N,N), B(N)
30 FOR I=1 TO N
40 FOR J=1 TO N
50 PRINT "a(";I;" ";J;")=";
60 INPUT A(I,J)
70 NEXT J,I
80 FOR I=1 TO N
90 FOR J=2 TO N
100 IF A(I,J-1)<A(I,J) THEN 120
110 B(I)=1: GOTO 140
120 NEXT J
130 B(I)=0
140 NEXT I

```

```

150 FOR I=1 TO N
160 PRINT "b(";I;")=";B(I);" ";
170 NEXT I

```

8. $A[1,N]$ ва $B[1,M]$ камаймайдиган тартибда сараланган иккита жадвалдан камаймайдиган тартибда сараланган $C[1,N+M]$ жадвал ҳосил қилувчи дастур тузинг.

Ечиш: Ушбу масалада бир-бирига боғланган учта такрорлануви 140-210, 220-250, 260-290 сатрлар мавжуд. Улар ёрдамида сараланган жадвал элементларини учинчи жадвалга жойлаштиришнинг барча ҳолатлари ҳисобга олинган.

```

10 INPUT "Elementlar soni N,M ";N,M
20 IF N<1 OR M<1 THEN 10
30 DIM A(N),B(M),C(N+M)
40 FOR I=1 TO N
50 PRINT "a(";I;")=";
60 INPUT A(I)
70 NEXT I
80 FOR I=1 TO M
90 PRINT "b(";I;")=";
100 INPUT B(I)
110 NEXT I
120 I=1
130 J=1
140 WHILE I<=N AND J<=M
150 IF A(I)<B(J) THEN 190
160 C(I+J-1)=B(J)
170 J=J+1
180 GOTO 210
190 C(I+J-1)=A(I)
200 I=I+1
210 WEND
220 IF I>N THEN 260
230 C(N+I)=A(I)
240 I=I+1
250 GOTO 220
260 IF J>M THEN 300
270 C(N+J)=B(J)
280 J=J+1
290 GOTO 260
300 FOR I=1 TO N+M
310 PRINT"C(";I;")=";C(I)
320 NEXT I
330 END

```

9. $A[1,N]$ ва $B[1,M]$ ўсмайдиган тартибда сараланган иккита жадвалдан ўсмайдиган тартибда сараланган $C[1,N+M]$ жадвал ҳосил қилувчи дастур тузинг.

Ечиш: Аввалги масала шартига тескари шарт қўйилган. Дастур тузиб кўриринг.

IV. Кетма-кет яқинлашиш

1. Куйидаги шартларни қаноатлантирувчи сонли кетма-кетликни экранга чиқаринг:

- кетма-кетликнинг биринчи ҳади 3 дан 97 гача бўлган ихтиёрий туб сон;
- кейинги ҳар бир ҳади ўзидан олдин келган ҳадга боғлиқ бўлиб, агар аввалги ҳад тоқ бўлса, $3p+1$ га, жуфт бўлса $p/2$ га тенг

Ҳисоблашни кетма-кетликда 1 сони ҳосил бўлгунча давом эттиринг.

Масалан: 17 52 26 13 40 20 10 5 16 8 4 2 1

Ечиш: Масалани ечишда аввало киритилган P сонини туб ёки туб эмаслигини аниқлаш зарур. Берилган сон туб бўлиши учун уни 2 да бошлаб то \sqrt{P} нинг бутун қисмигача бўлган сонларга кетма-кет бўлиш зарур. Агар P сони шу сонларнинг бирортасига ҳам бўлинмаса, яъни $P/I = \text{INT}(P/I)$ шарт бажарилмаса P сони туб бўлади. Берилган P сони учун $P/2 = \text{INT}(P/2)$ шарт бажарилса P сони жуфт, акс ҳолда тоқ бўлади.

```
10 INPUT "Tub sonni kiriting ";P
20 PRINT P;
30 FOR I=2 TO INT(SQR(P))
40 IF P/I=INT(P/I) THEN PRINT"Kiritilgan ";P;" soni tub emas":GOTO 10
50 NEXT I
60 IF P/2=INT(P/2) THEN P=P/2: GOTO 80
70 P=P*3+1
80 PRINT P;
90 IF P=1 THEN 110 ELSE 60
100 END
RUN
```

2. a , x , s ҳақиқий сонлар берилган. $Y_0 = a$; $Y_i = 1/2(Y_{i-1} + X/Y_{i-1})$, $i=1,2,\dots$ = онуният асосида ҳосил қилинган Y_1, Y_2, \dots кетма -кетликнинг $|Y_n^2 - Y_{n-1}^2| < S$ шартни қаноатлантирувчи биринчи Y_n - ҳадни топиш дастурини тузинг.

Ечиш: Масалани ечишда кетма-кетликни ҳосил қилувчи боғланишни тасаввур қила билиш талаб қилинади. $Y_i = 1/2(Y_{i-1} + X/Y_{i-1})$ кетма-кетликнинг битта ҳади бўлиб унинг биринчи ҳади $Y_0 = a$ га, иккинчи ҳади $Y_1 = 1/2(a + X/a)$ га тенг. Учинчи ҳадини ҳисоблашда Y_{i-1} ларнинг ўрнига Y_1 нинг қиймати қўйилади ва ҳ.к.

```
10 INPUT "a=";A
20 INPUT "x=";X
30 INPUT "s=";S
40 Y0=A
50 Y1=(Y0+X/Y0)/2
60 IF ABS(Y1-Y0)<S THEN 80
70 Y0=Y1: GOTO 50
80 PRINT Y1
90 END
```

3. a , m , x , s ҳақиқий сонлар берилган. $Y_0 = a$; $Y_i = \frac{m-1}{m}Y_{i-1} + \frac{x-1}{m+1}Y$, $i=1,2,\dots$ = онуният асосида ҳосил қилинган Y_1, Y_2, \dots кетма-кетликнинг $|Y_n - Y_{n-1}| < s$ шартини қаноатлантирувчи биринчи Y_n -ҳадни топиш дастурини тузинг.

Ечиш: Аввалги масала шартига ўхшаш.

```
10 INPUT "a=";A
20 INPUT "m=";M
30 INPUT "x=";X
40 INPUT "s=";S
50 Y0=A
60 Y1=((M-1)/M)*Y0-(X-1)/(M+1)*Y0
70 IF ABS(Y1-Y0)<S THEN 90
80 Y0=Y1: GOTO 60
90 PRINT Y1
100 END
```

V. Максимум ва минимумларга доир масалалар

1. Натурал k, n ва ҳақиқий a_1, \dots, a_{kn} сонлар берилган. $\max(a_1, \dots, a_k)$, $\max(a_{k+1}, \dots, a_{2k})$, $\max(a_{k(n-1)}, \dots, a_{kn})$ кетма-кетлик ҳосил бўлувчи дастур тузинг.

Ечиш. Ҳақиқий сонлар кетма-кетлиги учта қисмга ажратилиб, уларнинг максимум қийматларини топиш талаб қилинганлиги учун берилган чегаралар бўйича алоҳида максимумлар топилган. Дастур тузишда кетма-кетликнинг максимумларини топиш учун қисм дастурдан фойдаланиш мақсадга мувофиқ.

```
10 INPUT K,N
20 DIM A(K*N)
30 FOR I=1 TO K*N
40 PRINT "A(";I;")="";INPUT A(I)
50 NEXT I
60 M1=A(1)
90 FOR I=2 TO K
100 IF M1<A(I) THEN M1=A(I)
110 NEXT I
120 M2=A(K+1)
130 FOR I=K+2 TO 2*K
140 IF M2<A(I) THEN M2=A(I)
150 NEXT I
160 M3=A(K*(N-1))
170 FOR I=K*(N-1)+1 TO K*N
180 IF M3<A(I) THEN M3=A(I)
190 NEXT I
200 PRINT M1,M2,M3
```

2. Натурал k, n ва ҳақиқий a_1, \dots, a_{kn} сонлар берилган. $\min(\max(a_1, \dots, a_k), \max(a_{k+1}, \dots, a_{2k}), \max(a_{k(n-1)+1}, \dots, a_{kn}))$ сонни ҳосил қилувчи дастур тузинг.

Ечиш: Дастур алгоритмининг тузишда кетма-кетликнинг учта қисмининг максимумларини топиш талаб қилинганлиги учун қисм дастурдан фойдаланиш мақсадга мувофиқ. +исм дастурда берилган кетма-кетликнинг максимуми топилади, сўнгра уларнинг минимуми аниқланади.

```
10 INPUT "k,n ";K,N
20 DIM A(K*N)
30 FOR I=1 TO K*N
40 PRINT "A(";I;")="";
50 INPUT A(I)
60 NEXT I
70 M=A(1):B=1:O=K:GOSUB 200
80 S1=M
90 M=A(K+1):B=K+1:O=2*K:GOSUB 200
100 S2=M
110 IF S1<S2 THEN MINIMUM=S1 ELSE MINIMUM=S2
120 M=A(k*(n-1)+1):B=K*(n-1)+1:O=K*N:GOSUB 200
130 S3=M
140 IF MINIMUM>S3 THEN MINIMUM=S3
150 PRINT "Minimumi =" ;MINIMUM
160 END
```

```
200 'Qism dastur
210 FOR I=B+1 TO O
220 IF M<A(I) THEN M=A(I)
230 NEXT I
240 RETURN
```

VI. Геометрик масалалар

1. Тўғри тўртбурчакнинг юзи q га тенг. Агар тўғри тўртбурчакнинг томонлари натурал сонларга тенг бўлса, юзи q га тенг бўлган барча тўғри тўртбурчакларнинг томонларини аниқлаш дастури тузилсин.

Ечиш: Тўғри тўртбурчак юзини q орқали ва унинг томонлари узунликларини a ва b лар орқали белгилайлик. А ниг бошлангич қиймати 1 га тенг қилиб оламиз ва a бўйича цикл очамиз. $a*b=q$ шартни қаноатлантирувчи барча a ($a \leq b$) ва b ларнинг бутун қийматларини топамиз.

Масалан, $q = 20$ бўлса, натижа 1 20, 2 10, 4 5 каби бўлади.

```
10 INPUT "q=";Q
20 A=1: B=Q
30 WHILE A<=B
40 IF A*B=Q THEN PRINT"a=";A,"b=";B
50 A=A+1
60 B=INT(Q/A)
70 WEND
80 END
```


2. Шундай a ва b ($|a| < 100$, $|b| < 100$) сонлар берилганки, $y=ax+b$ тўғри чизик координаталар ўқи билан биргаликда чегараланган учбурчак ҳосил қилади. Учбурчак координаталар текислигининг қайси чорагида ётишини аниқловчи дастур тузинг.

Ечиш. Учбурчак координаталар текислигининг I чорагида ётиши учун координата ўқларини кесиб ўтувчи чизик абсциссаси ва ординатаси мусбат, яъни $-b/a > 0$, $b > 0$ ва $a \neq 0$ бўлиши талаб қилинади. II чорракда ётиши учун x ўқи абсциссаси манфий, ординатаси мусбат, яъни $-b/a < 0$, $b > 0$ ва $a \neq 0$ бўлиши талаб қилинади ва ҳ.к.

```
10 INPUT "a="; A
20 INPUT "b="; B
30 IF ABS(A)>=100 OR ABS(B)>=0 OR A=0 THEN PRINT "Координаталар нотугри киритилган.": GOTO 10
40 IF -B/A>0 AND B>0 AND A<>0 THEN PRINT "Учбурчак биринчи чоракда ётади"
50 IF -B/A<0 AND B>0 AND A<>0 THEN PRINT "Учбурчак иккинчи чоракда ётади"
60 IF -B/A<0 AND B<0 AND A<>0 THEN PRINT "Учбурчак учинчи чоракда ётади"
70 IF -B/A>0 AND B<0 AND A<>0 THEN PRINT "Учбурчак тўртинчи чоракда ётади"
80 END
```

3. Томонлари A , B натурал сонга тенг бўлган тўғри тўртбурчак берилган. Агар ундан ҳар сафар томонлари натурал сонларга тенг бўлган энг катта квадрат кесиб олинаверса, нечта квадрат ҳосил бўлишини аниқловчи дастур тузинг.

Ечиш: Масалани ечишда ҳар сафар катта томондан кичик томонни қирқиб олиш, ва ҳар бир қирқиб олинганда қирқиб олинган квадратлар сонини ҳисоблаб туриш лозим.



```
10 'Kvadrat
20 INPUT "A,B larni kiriting:";A,B
30 S=0
40 IF A=B THEN S=S+1: GOTO 120
```

```

50 IF A>B THEN 90
60 B=B-A
70 S=S+1
80 GOTO 40
90 A=A-B
100 S=S+1
110 GOTO 40
120 PRINT "Kvadratlar soni: ";S;" ta"
130 END
RUN
A,B larni kiriting: ? 4,11
Kvadratlar soni: 6 ta
Ok
RUN
A,B larni kiriting: ? 12,8
Kvadratlar soni: 3 ta

```


4. Учбурчак ўзининг (x_1, y_1) , (x_2, y_2) , (x_3, y_3) учларининг координаталари ва нукта (x, y) координатаси билан берилган. Берилган нуктанинг учбурчак ичида ётиши ёки ётмаслигини аниқланг.

Ечиш: Нуктанинг учбурчак ичида ётишини текшириш учун уни учбурчак учлари билан туташтириб ҳосил бўлган учбурчак юзалари ҳисобланади. Ҳосил бўлган учбурчаклар юзаларининг йиғиндиси берилган учбурчак юзига тенг бўлса нукта учбурчак ичида ётади, акс ҳолда нукта учбурчак ичида ётмайди.

Учбурчак юзасини ҳисоблашда $S = \sqrt{p(p-a)(p-b)(p-c)}$, бунда $p = \frac{a+b+c}{2}$ Герон

формуласидан ёки $S = \frac{1}{2} |(x_2 - x_1)(y_3 - y_1) - (x_3 - x_1)(y_2 - y_1)|$ формуладан фойдаланиш мумкин.

Ушбу мисолимизда иккинчи формуладан фойдаланиш мақсадга мувофиқ.



5. Текисликда берилган нукталар тўплами ичидан шундай учтасини топингки, улар энг катта юзали учбурчак ҳосил қилсин.

Ечиш: Ушбу масаланинг дастурини икки хил усулда, яъни учбурчак юзини ҳисоблашда Герон формуласидан ва $T = \text{ABS}((X(2)-X(1))*(Y(3)-Y(1))-(X(3)-X(1))*(Y(2)-Y(1)))/2$ формуладан фойдаланиб тузиш кўрсатилган. Шунингдек, иккала ҳолда ҳам бир хил натижа олинганлиги тасдиқланган. Хулоса қилиб айтганда, ўзининг координаталари билан берилган учбурчакнинг юзини ҳисоблашда иккинчи формуладан фойдаланиш мақсадга мувофиқ экан.

Текисликдаги берилган N та нуқтадан учтадан группаласак барча $C_N^3 = \frac{N!}{3!(n-3)!}$ та

учбурчаклар ҳосил бўлади. Масалан, текисликда 5 та нуқта берилган бўлса, уларни мос равишда 1,2,3,4,5 деб белгилайлик. Уларни учтадан группалаганда

$C_5^3 = \frac{5!}{3!(5-3)!} = \frac{4 \cdot 5}{2} = 10$ та яъни (1,2,3), (1,2,4), (1,2,5), (1,3,4), (1,3,5), (1,4,5), (2,3,4),

(2,3,5), (2,4,5), (3,4,5) учларига эга бўлган учбурчаклар ҳосил бўлади. Шунинг учун циклини FOR I=1 TO N, FOR J=I+1 TO N, FOR K=J+1 TO N кўринишда очиш мақсадга мувофиқ.

I - дастур.

```

10 INPUT "Tekislikdagi nuqtalar soni=";N
20 DIM X(N), Y(N)
30 FOR I=1 TO N
40 PRINT"x(";I;")=";
50 INPUT X(I)
60 PRINT"y(";I;")=";
70 INPUT Y(I)
80 NEXT I
90 S=0
100 A=SQR((X(1)-X(2))^2+(Y(1)-Y(2))^2)
110 B=SQR((X(2)-X(3))^2+(Y(2)-Y(3))^2)
120 C=SQR((X(3)-X(1))^2+(Y(3)-Y(1))^2)
130 P=(A+B+C)/2
140 T=SQR(P*(P-A)*(P-B)*(P-C))
150 FOR I=1 TO N
160 FOR J=I+1 TO N
170 FOR K=J+1 TO N
180 A=SQR((X(I)-X(J))^2+(Y(I)-Y(J))^2)
190 B=SQR((X(J)-X(K))^2+(Y(J)-Y(K))^2)
200 C=SQR((X(K)-X(I))^2+(Y(K)-Y(I))^2)
210 P=(A+B+C)/2
220 S=SQR(P*(P-A)*(P-B)*(P-C))
230 IF T>S THEN 260
240 T=S:D=I:E=J:F=K
250 NEXT K
260 NEXT J
270 NEXT I
280 PRINT"Eng katta yusali uchburchak ysasi ";T;" ga teng"
290 PRINT "Uning koordinatalari A(";X(D);";";Y(D);") ";
300 PRINT "B(";X(E);";";Y(E);") "; "C(";X(F);";";Y(F);")"
310 END

```

II – дастур.

```

10 INPUT "Tekislikdagi nuqtalar soni=";N
20 DIM X(N), Y(N)
30 FOR I=1 TO N
40 PRINT"x(";I;")=";
50 INPUT X(I)
60 PRINT"y(";I;")=";
70 INPUT Y(I)
80 NEXT I
90 S=0

```

```

100 T=ABS((X(2)-X(1))*(Y(3)-Y(1))-(X(3)-X(1))*(Y(2)-Y(1)))/2
110 FOR I=1 TO N
120 FOR J=I+1 TO N
130 FOR K=J+1 TO N
140 S=ABS((X(I)-X(J))*(Y(K)-Y(J))-(X(K)-X(J))*(Y(I)-Y(J)))/2
150 IF T>S THEN 250
160 T=S:D=I:E=J:F=K
170 NEXT K
180 NEXT J
190 NEXT I
200 PRINT"Eng katta yusali uchburchak ysasi ";T;" ga teng"
210 PRINT "Uning koordinatalari A(";X(D);";";Y(D);") ";
220 PRINT "B(";X(E);";";Y(E);") "; "C(";X(F);";";Y(F);")"
230 END

```

6. Текисликда берилган нуқталар тўплами ичидан шундай учтасини топингки, улар энг кичик юзали учбурчак ташкил қилсин.

Ечиш: Ушбу масалани ечишда ҳам аввалги масала шартларидан фойдаланиш мумкин. Дастур тузиб кўйринг.

7. Текисликда берилган нуқталар ичидан шундай нуқтани топингки, қолган нуқталаргача бўлган масофалар йиғиндиси энг кичик бўлсин.

Ечиш: Текисликдаги икки нуқта орасидаги масофа $d = \sqrt{(x_2 - x_1)^2 + (y_2 - y_1)^2}$ формула ёрдамида ҳисобланади. Ҳар бир нуқтадан қолган нуқталаргача бўлган масофалар кўшилиб янги $B[1, N]$ жадвал орқали ифодаланади. Янги жадвал элементларининг минимуми топилади.

```

10 INPUT "n=";N
20 DIM A(N),B(N)
30 FOR I=1 TO N
40 PRINT "x(";I;")=";
50 INPUT X(I)
60 PRINT "y(";I;")=";
70 INPUT Y(I)
80 NEXT I
90 FOR I=1 TO N
100 K=0
110 FOR J=1 TO N
120 IF I=J THEN 140
130 K=K+SQR((X(I)-X(J))^2+(Y(I)-Y(J))^2)
140 NEXT J
150 B(I)=K
160 L=I
170 NEXT I
180 T=B(1)
190 FOR I=1 TO N
200 IF T<B(I) THEN 230
210 T=B(I)
220 L=I
230 NEXT I
240 PRINT "X(";L;")=";X(L);"Y(";L;")=";Y(L);" nuqtadan qolgan";

```


```

250 PRINT " nuqtalargacha bo'lgan masofalar yigindisi eng kichik"
260 PRINT "Eng kichik masofa=";T
270 END

```

8. $X_1, Y_1, X_2, Y_2, \dots, X_8, Y_8$ хақиқий сонлар берилган. $(X_1, Y_1), (X_2, Y_2), (X_3, Y_3), (X_4, Y_4)$ координатали нукталар биринчи тўғри бурчакли тўртбурчакнинг учлари, $(X_5, Y_5), (X_6, Y_6), (X_7, Y_7), (X_8, Y_8)$ нукталар иккинчи тўғри бурчакли тўртбурчакнинг учлари деб қаралади. Иккинчи тўртбурчак биринчиси ичида ётишини аниқланг, агар ётса биринчи тўртбурчакнинг иккинчи тўртбурчакка тегишли эмас қисми юзасини топинг.

Ечиш: Берилган нукта қабарик кўпбурчак ичида ётишини текшириш учун шу нукта билан



қабарик кўпбурчак учлари туташтирилади. Ҳосил бўлган учбурчаклар йиғиндиси қабарик кўпбурчак йиғиндисига тенг бўлса, нукта шу кўпбурчак ичида ётади, акс ҳолда нукта кўпбурчак ташқарисидида ётади. Ю=оридаги мулоҳазага асосан, иккинчи тўртбурчакнинг ҳар бир учи кетма-кет биринчи тўртбурчак ичида ётиши текширилади.

Учларининг координаталари берилган учбурчакнинг юзини ҳисоблашда Герон формуласидан фойдаланилади. Бизнинг мисолимизда учбурчак юзини ҳисоблашда

$S = \frac{1}{2} |(x_2 - x_1)(y_3 - y_1) - (x_3 - x_1)(y_2 - y_1)|$ формуладан фойдаланилди.

```

10 DIM X(8),Y(8)
20 FOR I=1 TO 8
30 PRINT "x(";I;")=";
40 INPUT X(I)
50 PRINT "y(";I;")=";
60 INPUT Y(I)
70 NEXT I
80 S1=SQR((X(1)-X(2))^2+(Y(1)-Y(2))^2)*SQR((X(1)-X(4))^2+(Y(1)-Y(4))^2)
90 S2=SQR((X(5)-X(6))^2+(Y(5)-Y(6))^2)*SQR((X(5)-X(8))^2+(Y(5)-Y(8))^2)
100 I=5:K=0
110 X=X(I):Y=Y(I)
120 GOSUB 200
130 IF S1=SU1+SU2+SU3+SU4 THEN K=K+1 ELSE 180
140 IF K<>4 THEN 170
150 PRINT "2-to'rtburchak 1-to'rtburchak ichida yotadi"
160 PRINT "To'rtburchak yusalarining ayirmasi ";S1-S2:GOTO 190
170 I=I+1:IF I<=8 THEN 110
180 PRINT "2-to'rtburchak 1-to'rtburchak ichida yotmaydi"
190 END
200 'Qism dastur
210 SU1=ABS((X(2)-X(1))*(Y-Y(1))-(X-X(1))*(Y(2)-Y(1)))/2
220 SU2=ABS((X(3)-X(2))*(Y-Y(2))-(X-X(2))*(Y(3)-Y(2)))/2
230 SU3=ABS((X(4)-X(3))*(Y-Y(3))-(X-X(3))*(Y(4)-Y(3)))/2
240 SU4=ABS((X(4)-X(1))*(Y-Y(1))-(X-X(1))*(Y(4)-Y(1)))/2
250 RETURN

```

9. $X_1, Y_1, X_2, Y_2, \dots, X_7, Y_7$ хақиқий сонлар берилган. $(X_1, Y_1), (X_2, Y_2), (X_3, Y_3), (X_4, Y_4)$ координатали нуқталар биринчи тўғри бурчакли тўртбурчакнинг учлари, $(X_5, Y_5), (X_6, Y_6), (X_7, Y_7)$ нуқталар учбурчакнинг учлари деб қаралади. Учбурчак тўртбурчакнинг ичида ётишини аниқланг, агар ётса тўртбурчакнинг учбурчакка тегишли эмас қисми юзасини топинг.

Ечиш: Аввалги масаладаги формулалардан фойдаланиш тавсия қилинади. Дастур тузиб кўририг.

VII. Белгили катталиклар билан ишлаш функцияларига оид масалалар

1. S сатр фақат лотин ҳарфларидан иборат. Сатрда қатнашган “а” белгиси турган ўринлари йиғиндисини ҳисобловчи дастур тузинг.

Ечиш. Маълумки, сатр узунлиги 255 та белгидан ошмайди. Сатрда фақат кичик “а” ҳарфларининг ўринларини ҳисобга олиш талаб қилинади. Дастур тузиб кўринг.

2. Қавслардан иборат сатр берилган. қавслар тўғри ёки нотўғри қўйилганини текшириб жавоб чиқарувчи дастур тузинг.

Масалан ((()))-тўғри
(())()-нотўғри

Ечиш: Очилган қавслар сонини S ва ёпилган қавсларни сонини эса T орқали белгилаб оламиз. қавсларни текширишнинг ҳар бир қадамида $S \geq T$ шартининг бажарилишини шарт. Акс ҳолда қавслар нотўғри қўйилган бўлади.

```
10 INPUT "Faqat qavslardan iborat satrni kiriting x$=";X$
20 S=0:T=0
30 FOR I=1 TO LEN(X$)
40 IF MID$(X$,I,1)="(" THEN S=S+1 ELSE T=T+1
50 IF S<T THEN 90
60 NEXT I
70 IF S=T THEN PRINT"Qavslar to'g'ri qo'yilgan"
80 GOTO 100
90 PRINT"Qavslar noto'g'ri qo'yilgan"
100 END
```

3. N натурал сонигача бўлган ўзининг барча рақамларига қолдиксиз бўлинувчи сонларни экранга чиқариш дастури тузилсин.

Ечиш: Берилган сонни сатр катталигига айлантирувчи STR\$ функциясини ишлатганда у сатр катталик олдига бўш жой белгисини қўшиб қўяди. Шунинг учун сатрга айлантирилган натурал соннинг узунлигини аниқлашда ундан 1 ни айириб қўйиш зарур.

Масалани ечишда N натурал сонигача бўлган ҳар бир соннинг рақамлари кетма-кет қирқиб олинади (60 сатр) ва қирқиб олинган сонни қайта сонга айлантириб аниқланаётган сонга бўлинади(80 сатр). Аниқланаётган соннинг ҳар бир рақами шу сонга бўлинса у экранга чиқарилади.

```
10 INPUT "n=";N
20 FOR I=1 TO N
30 K$=STR$(I)
40 T=LEN(K$)-1
50 FOR J=1 TO T
60 X$=MID$(STR$(I),J+1,1)
70 IF X$="0" THEN 90
80 IF I/VAL(X$)<>INT(I/VAL(X$)) THEN 110
90 NEXT J
100 PRINT I;
110 NEXT I
120 END
RUN
```

4. Берилган сатрда қатнашган барча натурал сонлар йиғиндисини ҳисобланг. Масалан, “ytrfd6,tu89uu,,,91,12hgf” сатрда қатнашган натурал сонлар учун $6+8+9+9+1+1+2=36$.

Ечиш: Бейсик трансляторида 0 да 9 гача бўлган рақамларнинг ўнли коди кетма-кет ўсиб бориш тартибида жойлашган бўлади. Шунинг учун берилган матн ичидаги рақамларни аниқлаш учун ҳар бир белги қирқиб олиниб уни ASC(“0”) дан ASC(“9”) гача ётиши текширилади. Шу ораликда ётган рақамлар бир-бирига қўшиб чиқилади.

```
10 INPUT "Matnni kiriting ";A$
20 S=0
30 FOR I=1 TO LEN(A$)
40 IF ASC(MID$(A$,I,1))<ASC("0") AND ASC(MID$(A$,I,1))>ASC("9") THEN 60
50 S=S+VAL(MID$(A$,I,1))
60 NEXT I
70 PRINT "Matndagi raqamlar yigindisi ";S;" ga teng"
80 END
```

5. Иккилик соноқ системасида берилган сонни ўн асосли бутун сонга ўтказинг.

Ечиш: Икки асосли санок системасидаги сонни $(a_0a_1a_2 \dots a_{n-1}a_n)_2 = a_02^n + a_12^{n-1} + a_22^{n-2} + \dots + a_{n-1}2^1 + a_n2^0$ формула ёрдамида ўн асосли санок системасига ўтказиш мумкин бўлганлиги учун икки асосли санок системасидаги ҳар бир сон 2 нинг даражалари бўйича ёйиб, натижалар бир-бирига қўшиб чиқилади.

```
10 INPUT "Ikkilik sanoq sistemasidagi sonni kiriting ";B$
20 L$="1":K$="0":F=LEN(B$):S=0:T=F
30 FOR I=1 TO T
40 C$=MID$(B$,I,1)
50 IF C$<>L$ AND C$<>K$ THEN 10
60 A=VAL(C$): S=S+A*2^(T-1):T=T-1
70 NEXT I
80 PRINT"S=";S
90 END
```

6. Берилган N учун рақамларининг йиғиндисини N га тенг ва N га бўлинадиган энг кичик сонни топиш дастури тузилсин. ($N < 1000$).

Ечиш: Берилган N сонига қаррали бўлган сонларнинг рақамлари йиғиндисини шу сон билан солиштирилади. Йиғинди берилган сонга тенг бўлса шу сон экранга чиқарилади, акс ҳолда N га қаррали бўлган соннинг навбатдагиси текширилади.

```
10 INPUT "Berilgan sonni kiriting ";N
20 IF N<0 OR N>999 THEN 10
30 I=2
40 A=I*N
50 B$=MID$(STR$(A),2)
60 S=0
70 FOR J=1 TO LEN(B$)
80 K$=MID$(B$,J,1)
90 S=S+VAL(K$)
100 NEXT J
110 IF N=S THEN 140
```

```

120 I=I+1
130 GOTO 40
140 PRINT N;" ga bo'linadigan eng kichik son ";A;" ga teng"
150 END

```

7. Символлар кетма-кетлиги берилган (сатрли). Сўзнинг матни рақамли символлар ташкил топган. Берилган сўзлардан сўз таркибида симметрик қисми мавжуд ва бу қисмнинг узунлиги ушбу сўз узунлигининг ярмидан катта бўлган сўзларни топинг ва чиқаринг.

Масалан: 534565411 сўзи топилди, симметрик қисми 45654.

Ечиш: Аввало рақамли символлардан ташкил топган сўздаги бўш жой (пробел) белгиси L қидирилади (60 сатр). Матндаги бўш жой позицияси аниқлангандан сўнг берилган матндан бўш жойгача бўлган қисми қирқиб олинади (70 сатр). Қириб олинган қисм симметриклиги текширилади ва симметрик қисмидаги белгилар сони K аниқланади (80-120 сатрлар). K жуфт бўлса, симметрик элементлар сони $2 \cdot K$ га, акс ҳолда $2 \cdot K - 1$ га тенг бўлади.

Матндаги охириги сўзни аниқлаш 40 сатр орқали амалга оширилади. Агар матн битта сўздан иборат бўлса уни 50 сатр текширади.

```

10 INPUT X$
20 L=0:Z=1:C$=" "
30 FOR I=1 TO LEN(X$)
40 IF I=LEN(X$) AND Z=I AND L>0 THEN A$=MID$(X$,L+1,I-L): GOTO 80
50 IF I=LEN(X$) AND Z=I THEN A$=X$: GOTO 80
60 IF MID$(X$,I,1)<>C$ THEN 150
70 A$=MID$(X$,L+1,I-(L+1)):L=I
80 N=LEN(A$):K=0
90 FOR J=1 TO INT(N/2)+1
100 IF MID$(A$,J,1)<>MID$(A$,N-J+1,1) THEN K=0: GOTO 120
110 K=K+1
120 NEXT J
130 IF N/2=INT(N/2) THEN K=2*K ELSE K=2*K-1
140 IF K>INT(N/2) THEN PRINT A$
150 Z=Z+1
160 NEXT I
170 END

```

8. Чапдан ҳам ўнгдан ҳам бир хил ўқиладиган сонлар полиндром сонлар дейилади. 25000 дан кичик тоқ полиндром сонларни экранга чиқарувчи дастур тузинг.

Ечиш: Масала шартига кўра тоқ полиндром сонларни қидириш сўралганлиги учун цикл параметри I 1 дан бошлаб тоқ сонларни қабул қилади. Ҳар бир тоқ сон қирқиб олиниб (S\$), унинг узунлиги (L) аниқланади. Қирқиб олинган сўз тескарисидан ёзилади (60-90 сатрлар) ва у сонга айлантрилади (100 сатр). Қирқиб олинган тоқ сон тескариланган сонга тенг бўлса, бу сон полиндром сон бўлади, акс ҳолда навбатдаги сон текширилади.

```

10 Polindrom
20 PRINT "Polindrom sonlar:"
30 FOR I=1 TO 25000 STEP 2
40 S$=STR$(I)
50 L=LEN(S$)
60 P$=""
70 FOR J=1 TO L

```

```

80 P$=MID$(S$,J,1)+P$
90 NEXT J
100 K=VAL(P$)
110 IF K=I THEN PRINT K;" ";
120 NEXT I
130 END

```

9. Агар N , M натурал сонлар бўлсин. Рақамларининг йиғиндиларининг квадрати M га тенг бўлган барча N дан кичик натурал сонларни аниқлаш дастури тузилсин.

Ечиш: 1 дан бошлаб $N-1$ гача бўлган барча сонлар кетма-кет сатр катталиikka (50 қатор) айлантрилади. Ҳар бир сатр рақамлари бирин кетин қиркиб олиниб сонга айлантрилади ва бир бирига қўшилади (60-90 қаторлар). Йиғинди квадратга кўтарилади (100 қатор) ва M сони билан солиштирилади (140 қатор). Агар улар бир-бирига тенг бўлса, экранга чиқарилади (130 қатор), акс ҳолда санагич S тозаланади ва цикл давом эттирилади (140-150 сатрлар). Бу шартлар бажарилмаса, “Bunday son yo’q” жавоби олинади (160 сатр).

```

10 INPUT "N=";N
20 INPUT "M=";M
30 S=0: S1=0
40 FOR I=1 TO N-1
50 K$=STR$(I)
60 FOR J=1 TO LEN(K$)
70 B$=MID$(K$,J,1)
80 S=S+VAL(B$)
90 NEXTJ
100 K=S^2
110 IF K<>M THEN 140
120 S1=S1+1
130 PRINT "I=";I
140 S=0
150 NEXT I
160 IF S1=0 THEN PRINT "Bunday son yo'q"
170 END

```

10. Квадрат илдизи бутун бўлган сонларни кетма-кет (пробелсиз) ёзилса, қуйидагича сатр ҳосил бўлади: 149162536496481... Шу сатрнинг n – ўрнида қандай рақам турганини аниқловчи дастур тузинг ($n < 20000$).

Ечиш: 1 дан бошлаб ҳар бир сон квадратга кўтарилади ва унинг рақамларининг сони ҳисобланиб борилади. Рақамлар сони изланаётган ўриндаги рақам позицияси рақамига тенг бўлса, шу сон экранга чиқарилади.

```

10 REM
20 INPUT "NECHANCHI O'RINDAGI SON";N
30 IF N<0 OR N>20000 THEN 20
40 T=0:I=1
50 K=I*I
60 L$=MID$(STR$(K),2)
70 FOR J=1 TO LEN(L$)
80 S$=MID$(L$,J,1):T=T+1

```

```

90 IF T=N THEN PRINT N;" - O'RINDAGI SON",S$: GOTO 110
100 NEXT J
110 I=I+1:GOTO 50
120 END

```

11. Рақамларининг йиғиндиси жуфт бўлган сонлар кетма-кет ёзилса қуйидаги кўринишдаги 2468111315171920222426... сонлар кетма-кетлиги ҳосил бўлади. Кетма кетликнинг n ҳади топилсин ($n < 200000$).

```

10 INPUT "N ni kiriting ";N
20 IF N<0 AND N>200000 THEN 10
30 L=0: I=2
40 M=0
50 K$=STR$(I)
60 FOR J=2 TO LEN(K$)
70 M=M+VAL(MID$(K$,J,1))
80 NEXT J
90 IF M/2=INT(M/2) THEN 100 ELSE 150
100 FOR J=2 TO LEN(K$)
110 Z$=MID$(K$,J,1)
120 L=L+1
130 IF L <> N THEN 140 ELSE 170
140 NEXT J
150 I=I+1
160 GOTO 40
170 PRINT "Ketma-ketlikdagi ";N;" chi son = "; Z$
180 END

```

12. Берилган матндаги ҳар хил ҳарфларни ва уларнинг сонини топиш дастурини тузинг. Катта ва кичик ҳарфлар фарқлансин. Пробел (бўш жой) белгиси ҳисобга олинмасин.

Ечиш: Ушбу масалани ечишда кейинги ҳар бир ҳад билан аввалги ҳадлар кетма-кет солиштирилади. Агар кейинги ҳад аввалги ҳадларга тенг бўлмаса йиғинди S га бир сони қўшилади, акс ҳолда текшириш давом этирилади (50-100 сатрлар). Матнда пробеллар катнашган бўлса (40-сатр) умумий йиғиндидан 1 сони айрилади (120-сатр).

```

10 INPUT "Matnni kiriting ";A$
20 S=0
30 FOR I=1 TO LEN(A$)
40 IF MID$(A$,I,1)=" " THEN 110
50 B$="yo'q"
60 J=1
70 WHILE J<I AND B$="yo'q"
80 IF MID$(A$,I,1)=MID$(A$,J,1) THEN B$="bor" ELSE J=J+1
90 WEND
100 IF B$="yo'q" THEN S=S+1:PRINT MID$(A$,I,1);" ";
110 NEXT I
130 PRINT:PRINT"Kiritilgan matndagi har xil xarflar soni ";S;" ta"
140 END
RUN

```

13. Берилган матндаги фақат бир марта учрайдиган белгиларни ва уларнинг сонини аниқланг.

Ечиш: Матндаги ҳар бир белги кетма-кет матндаги барча белгилар билан солиштирилади. Текшириладиган белги матн ичида фақат бир марта учраса, шу белги экранга чиқарилади ва ҳисоблагич Т га 1 сони қўшилади.

```
10 INPUT "Matnni kiriting ";T$
20 T=0
30 FOR I=1 TO LEN(T$)
40 K=0:A$=MID$(T$,I,1)
50 IF A$=" " THEN 100
60 FOR J=1 TO LEN(T$)
70 IF A$=MID$(T$,J,1) AND I<>J THEN K=K+1
80 NEXT J
90 IF K=0 THEN PRINT A$;" ";T=T+1
100 NEXT I
110 PRINT:PRINT "Har xil harflar soni ";T;" ta"
120 END
```


90 END

IX. Календар

1. Маълумки йилни 12 даврга ажратиб уларнинг ҳар бирини бурж белгиларига мос қўйилади. Масалан, 20.01-18.02 Далв, 19.02-20.03 Ҳут, 21.03-20.04 Хамал, 21.04-21.05 Савр, 22.05-21.06 Жавзо, 22.06-22.07 Саратон, 23.07-22.08 Асад, 23.08-23.09 Сумбула, 24.09-23.10 Мезон, 24.10-22.11 А=раб, 23.11-21.12 Қавс, 22.12-19.01 Жадй.

Кирилган кун ва ойга мос келувчи бурж белгисининг номини аниқлаш дастури тузилсин.

Ечиш: Масала ечими содда бўлиб, кирилган кун буржга мос келувчи ойнинг биринчи кунини солиштириш етарли.

```
10 DIM D$(13),N(13)
20 $(1)="JADY":D$(2)="DALV":D$(3)="XUT":D$(4)="XAMAL":D$(5)="SAVR"
30 D$(6)="JAVZO":D$(7)="SARATON":D$(8)="ASAD":D$(9)="SUNBULA"
40 D$(10)="MEZON":D$(11)="AKRAB":D$(12)="QAVS":D$(13)="JADY"
50 N(1)=20:N(2)=19:N(3)=21:N(4)=21:N(5)=22:N(6)=22:N(7)=23:N(8)=23
60 N(9)=24:N(10)=24:N(11)=23:N(12)=22:N(13)=20
70 INPUT "Tug'ilgdn kun va oyingizni kiriting (kk,oo) ";K,O
80 IF K<N(O) THEN PRINT "Sizning burjingiz ";D$(O) ELSE PRINT"Sizning burjingiz
";D$(O+1)
90 END
RUN
Tug'ilgdn kun va oyingizni kiriting (kk,oo) ? 2,04
Sizning burjingiz XAMAL
```

Адабиётлар рўйхати

1. С.А.Абрамов, Г.Г.Гнездилова, Е.Н.Капустина, М.И.Селюн. Задачи по программированию.-М.:Наука,Гл.ред.физ.-мат.лит.,1988.-224 с.

2. Т.Р.Азларов, Ш.Ш.Аширов. Информатикадан олимпиада масалаларини ечиш. Мактаб ва ўрта ўқув юртлари ўқитувчилари учун методик қўлланма.-Т.:1993.-78 б.